EBSZ gyakorló feladatok 2.

I. Képlékeny alakváltozás

1. feladat

Egy l = 2 m hosszú, állandó 50×20 mm^2 téglalap keresztmetszetű kéttámaszú rudat a rúd közepén a rúd hossztengelyére merőleges irányú (felfelé mutató) koncentrált erő terheli.
a) Mekkora F_F erő esetén indul meg a képlékeny alakváltozás?
b) Mekkora lesz a a rugalmas mag mérete 1,1 F_F erő esetén?
c) A rúd hossztengelye irányában milyen hosszú lesz a képlékeny zóna kiterjedése?
E = 206,8 GPa, σ_F = 225 MPa.

Megoldás

Adatok:

```
In[1]:= l = 2;
a = 0.05;
b = 0.02;
σF = 225 × 10^6;
```

A keresztmetszet másodrendű nyomatéka az y-tengelyre $[m^4]$

```
In[5]:= Iy = a b^3 / 12
Out[5]= 3.33333 × 10^-8
```

Az x-irányú rúdban a hajlítónyomatéki igénybevétel ebben az esetben (F_F +y irányú)

```
In[6]:= Mh[x_] := Piecewise[{{+ FF/2 x, x < 1/2},
    {+ FF/2 x - FF (x - 1/2), x ≥ 1/2}}]
```

```
In[7]:= Mh[x] /. FF → F // Simplify // TraditionalForm
```

Out[7]//TraditionalForm=

$$\begin{cases} \frac{F x}{2} & x < 1 \\ F - \frac{F x}{2} & \text{True} \end{cases}$$

Ha például F_F 1N nagyságú lenne, akkor így nézne ki az igénybevételi függvény:

In[8]:=
```
Plot[Mh[x] /. FF → 1, {x, 0, l}, Filling → Axis, AxesLabel → {"x", "Mh"}]
```

Out[8]=



A maximális hajlító igénybevétel $x = l/2$ helyen ébred

In[9]:=
```
MhMax = Mh[l / 2]
```

Out[9]= $\frac{FF}{2}$

a) Képlékeny alakváltozás megindulása

Navier-képletből:

In[10]:= $\sigma xMax = \frac{MhMax}{Iy} \frac{b}{2}$

Out[10]= 150000. FF

Ezt egyenlővé téve a folyáshatárral:

In[11]:=
```
moa = First@Solve[σxMax == σF]
```

Out[11]= {FF → 1500.}

b) Rugalmas mag + képlékeny zóna

Ebben az esetben a hajlító igénybevétel módosul:

In[12]:=
```
MhMaxB = MhMax /. FF → 1.1 FF
```

Out[12]= 0.55 FF

Azaz 825 Nm

In[13]:=
```
MhMaxB = MhMaxB /. moa
```

Out[13]= 825.

A rugalmas és képlékeny részek közti határra vonatkozó képlet (téglalap keresztmetszetre)

In[14]:= $zF = \sqrt{3\left(\frac{b^2}{4} - \frac{MhMaxB}{\sigma F\ a}\right)}$

Out[14]= 0.00894427

Tehát a rugalmas mag teljes hossza a keresztmetszetben:

In[15]:=
```
2 zF
```
Out[15]=
```
0.0178885
```

Azaz 17.89 mm

Megjegyzés: a hajlításból származó feszültség eloszlás ekkor:

In[16]:=
$$\sigma x[z\_] := \text{Piecewise}\left[\left\{\left\{\frac{\sigma F}{zF} z, \text{Abs}[z] < zF\right\}, \{\text{Sign}[z]\,\sigma F, \text{Abs}[z] \geq zF\}\right\}\right]$$

In[17]:=
```
Plot[σx[z], {z, -b / 2, b / 2}, AxesLabel → {"z", "σx(z)"}, Filling → Axis]
```
Out[17]=


Ellenőrizhetjük az eloszlást, a $M_h = \int_A z\,\sigma_x(z)\,dA$ összefüggéssel

In[18]:=
$$\int_{-a/2}^{a/2}\int_{-b/2}^{b/2} z\,\sigma x[z]\,dz\,dy$$
Out[18]=
```
825.
```

(Szintén 825 Nm jön ki, tehát helyes a $\sigma_x(z)$ eloszlás.)

c) Képlékeny zóna kiterjedése (hosszirányban)

Meg kell keresnünk azt a pontot, ahol a rugalmas és képlékeny zóna határa megegyezik a kereszt-metszet szélességével $z_F = b/2$, azaz nincs képlékeny zóna

In[19]:=
$$\text{moc} = \text{First@Solve}\left[\frac{b}{2} == \sqrt{3\left(\frac{b^2}{4} - \frac{Mh0}{\sigma F\, a}\right)}\right]$$
Out[19]=
```
{Mh0 → 750.}
```

Azaz ott ahol a hajlító igénybevétel lecsökken 750 Nm-re, már eltűnik a képlékeny zóna

Mivel a hajlító igénybevétel függvénye az első szakaszon $M_h = \frac{(1.1\,F_F)}{2}\,x$

In[20]:=
$$\text{mocx} = \text{First@Solve}\left[\frac{1.1\,FF}{2}\,x == 750\ /.\ \text{moa}\right]$$
Out[20]=
```
{x → 0.909091}
```

Ez a rúd bal oldalán levő csak-rugalmas alakváltozással rendelkező rész hosszát adja meg, tehát a csak rugalmas zónával rendelkező rész:

In[21]:=
```
lR = 2 x /. mocx
```
Out[21]= 1.81818

Azaz a képlékeny rész hosszirányú kiterjedése

In[22]:=
```
lF = l - lR
```
Out[22]= 0.181818

Azaz l_F = 181.8 mm

Megjegyzés I: z_F változása a hossz mentén

Az 1.1 F_F erőhöz tartozó hajlító igénybevétel:

In[23]:= MhB[x_] := Piecewise$\left[\left\{\left\{+\frac{1.1\,FF}{2}\,x,\ x<\frac{l}{2}\right\},\ \left\{+\frac{1.1\,FF}{2}\,x-1.1\,FF\left(x-\frac{l}{2}\right),\ x\geq\frac{l}{2}\right\}\right\}\right]$ /. moa

In[24]:=
```
Plot[MhB[x], {x, 0, l}, Filling → Axis, AxesLabel → {"x", "Mh"}]
```
Out[24]=



Ezt behelyettesítve a z_F-re vonatkozó képletbe:

In[25]:= zFB[x_] := Min$\left[\sqrt{3\left(\frac{b^2}{4}-\frac{\text{MhB}[x]}{\sigma F\ a}\right)},\ b/2\right]$

In[26]:=
```
Plot[zFB[x], {x, 0, 1}, Filling → Axis, PlotRange → {0, b / 2},
 AxesLabel → {"x", "zF"}, PlotLabel → "Rugalmas-képlékeny határ 1.1 FF esetén"]
```

Out[26]=

Azaz itt is látszik, hogy csak a belső ~10%-ban van képlékeny alakváltozás.
(A kék zóna rugalmas, a fehér zóna képlékeny)

Megjegyzés 2: z_F változása a hossz mentén, 1.48 F_F esetén

Téglalap keresztmetszetnél az alaktényező

In[27]:=

$$MhF = \frac{a\, b^2}{6} \sigma F;$$

$$MhK = \frac{a\, b^2}{4} \sigma F;$$

$$\lambda = \frac{MhK}{MhF}$$

Out[29]= 1.5

Azaz 1.48 F_F nagyon közel van a képlékeny tartalék kimerüléséhez:

In[30]:=

$$MhC[x\_] := Piecewise\left[\left\{\left\{+\frac{(1.48\, FF)}{2} x,\ x < \frac{1}{2}\right\},\ \left\{+\frac{(1.48\, FF)}{2} x - (1.48\, FF)\left(x - \frac{1}{2}\right),\ x \geq \frac{1}{2}\right\}\right\}\right] \;/.\; moa$$

$$zFC[x\_] := Min\left[\sqrt{3\left(\frac{b^2}{4} - \frac{MhC[x]}{\sigma F\, a}\right)},\ b / 2\right]$$

In[32]:=
```
Plot[zFC[x], {x, 0, l}, Filling → Axis, PlotRange → {0, b / 2},
  AxesLabel → {"x", "zF"}, PlotLabel → "Rugalmas-képlékeny határ 1.48 FF esetén"]
```

Out[32]=



2. feladat

In[33]:=
```
Clear[σx, Mh, MhF, MhK, σF]
```

Mekkora az alaktényezője annak a T-szelvénynek, amelynek a magassága 150 mm, szélessége 100 mm,
vastagsága pedig 12 mm?

In[34]:=
```
h = 150;
w = 100;
t = 12;
```

A keresztmetszet alakja, súlypontja, másodrendű nyomatéka

Bontsuk fel a T szelvényt két téglalapra:

In[37]:=
```
Graphics[{EdgeForm[Black], Lighter@Yellow,
  Rectangle[{-w / 2, h - t}, {w / 2, h}],
  Rectangle[{-t / 2, 0}, {t / 2, h - t}]
 }, Frame → True, GridLines → Automatic, FrameLabel → {"y", "z"}]
```

Out[37]=



Határozzuk meg a súlypontját:

In[38]:=
```
yS1 = 0;
zS1 = h - t / 2;
A1 = w * t;
yS2 = 0;
zS2 = (h - t) / 2;
A2 = (h - t) * t;
```

In[44]:= $yS = \frac{yS1\ A1 + yS2\ A2}{A1 + A2}$

Out[44]= 0

Megjegyzés: látható, hogy a súlypont x_S koordinátája 0.

In[45]:= $zS = \frac{zS1\ A1 + zS2\ A2}{A1 + A2}$ // N

Out[45]= 100.513

In[46]:=
```
Graphics[{EdgeForm[Black], Lighter@Yellow,
   Rectangle[{-w / 2, h - t}, {w / 2, h}],
   Rectangle[{-t / 2, 0}, {t / 2, h - t}],
   Black, PointSize[Medium], Point[{yS, zS}]
  }, Frame → True, GridLines → Automatic, FrameLabel → {"y", "z"}]
```

Out[46]=


A súlypontra azért van szükség, hogy a két téglalap másodrendű nyomatékát átszámíthassuk saját súlyponti tengelyeikről a közös súlyponton átmenő y tengelyre.

In[47]:= $It1y = \frac{w\ t^3}{12}$;

$It2y = \frac{t\ (h - t)^3}{12}$;

Az átszámítás steiner tétellel történik: $I_{yS} = I_{yS1} + A\,(z_S - z_{S1})^2$

In[49]:= $It1Sy = It1y + A1\ (zS1 - zS)^2$

Out[49]= 2.28378×10^6

In[50]:= `It2Sy = It2y + A2 (zS2 - zS)^2`

Out[50]= 4.27255×10^{6}

A keresztmetszet teljes Iy másodrendű nyomatéka a két téglalap másodrendű nyomatékainak összege. (Csak akkor lehet összeadni, ha már átszámítottuk a közös súlyponti tengelyre)

In[51]:= `ISy = It1Sy + It2Sy`

Out[51]= 6.55634×10^{6}

A folyás kezdetéhez tartozó hajlító igénybevétel

$\sigma_x = \frac{M_h}{I_y} z_{max}$ egyenletből

In[52]:= `zmax = Max[zS, h - zS]`

Out[52]= 100.513

Tehát az alsó szál kezd majd megfolyni

In[53]:= $MhF = \frac{\sigma F\ ISy}{zmax}$

Out[53]= 65 229. σF

A teljes megfolyáshoz tartozó igénybevétel

Ebben az esetben $\sigma_x = \text{sgn}(z)\, \sigma_F$

A keresztmetszet statikai nyomatéka három részből adódik:
Az első a fenti téglalaprész: $A_1\,(z_{S1} - z_S)$,
a másik kettő a középső rész közös súlypont feletti és alatti részei:
$A_{2+} = t * (h - t - z_S)$
$z_{2+} = (h - t - z_S)/2$
$A_{2-} = t * z_S$
$z_{2-} = z_S/2$

In[54]:= $Sy = A1\,(zS1 - zS) + t\,\frac{(h - t - zS)^2}{2} + t\,\frac{zS^2}{2}$

Out[54]= 117 017.

In[55]:= `MhK = Sy σF`

Out[55]= 117 017. σF

Az alaktényező

In[56]:= `λ = MhK / MhF`

Out[56]= 1.79395

Azaz $\lambda \sim 1.8$

3. feladat

In[57]:=
```
Clear[a, b, σF]
```

Egy 62,5 mm belső és 190 mm külső sugarú cső üzemi nyomása 240 MPa.
a) Mekkora a képlékeny folyással szembeni biztonsági tényező a Mohr-elmélet alapján?
b) Mekkora a képlekeny zóna sugara 580 MPa próbanyomás után?
c) Mekkora lesz a maximális tangenciális feszültség a csőben a próbaterhelés hatására?
d) Mekkora lesz az egyenértékű maradó feszültség a cső belső peremén a tehermentesítés után?
E = 200 GPa, σ_F = 850 MPa.

In[58]:=
```
a = 0.0625;
b = 0.19;
pb = 240 × 10^6;
pp = 580 × 10^6;
σF = 850 × 10^6;
pk = 0;
```

a) Folyással szembeni biztonsági tényező

A feszültségeloszlásban szereplő paraméterek:

In[64]:= $A = \frac{pb\, a^2 - pk\, b^2}{b^2 - a^2}$ // N

Out[64]= 2.91206×10^7

In[65]:= $B = (pb - pk) \frac{a^2 b^2}{b^2 - a^2}$ // N

Out[65]= 1.05125×10^6

Azaz a feszültségeloszlások

In[66]:= $\sigma r[r\_] := A - \frac{B}{r^2}$

$\sigma\theta[r\_] := A + \frac{B}{r^2}$

In[68]:=
```
Plot[{σr[r], σθ[r]}, {r, a, b}, PlotLegends → {"σr(r)", "σθ(r)"},
 AxesLabel → {"r", "σ"}, PlotRange → {{0, b}, All}, Filling → Axis]
```

Out[68]=



A Mohr-féle egyenértékű feszültség:

In[69]:=
```
σeMohrMax = σθ[a] - σr[a]
```
Out[69]= 5.38241×10^{8}

Azaz 538 MPa

A biztonsági tényező

In[70]:=
$$nF = \frac{\sigma F}{\sigma eMohrMax}$$
Out[70]= 1.57922

Azaz n ~ 1.58

b) Képlékeny zóna sugara

A képlékeny zóna sugarának és a belső nyomásnak a kapcsolatát a

$p = \frac{\sigma_F}{2}\left(\ln\left[\left(\frac{c}{a}\right)^2\right] + 1 - \left(\frac{c}{b}\right)^2\right)$

összefüggés adja meg.

Ebből az egyenletből c-t kifejezni csak numerikusan lehet, ha kirajzoljuk a

$\frac{\sigma_F}{2}\left(\ln\left[\left(\frac{c}{a}\right)^2\right] + 1 - \left(\frac{c}{b}\right)^2\right) - p$

függvényt, akkor annak x-tengelymetszete adja meg c értékét.

In[71]:=
$$\text{Plot}\left[\frac{\sigma F}{2}\left(\text{Log}\left[(c/a)^2\right] + 1 - (c/b)^2\right) - pp, \{c, a, b\}\right]$$

Out[71]=


In[72]:=
$$mob = \text{FindRoot}\left[\frac{\sigma F}{2}\left(\text{Log}\left[(c/a)^2\right] + 1 - (c/b)^2\right) - pp, \{c, a\}\right]$$
Out[72]= {c → 0.0823975}

In[73]:=
```
c = c /. mob
```
Out[73]= 0.0823975

Azaz c ~ 82.4 mm

Számológéppel ugyanez az ún. felező módszerrel tehető meg:
Első közelítésben célszerű a-t és (a+b)/2-t behelyettesíteni:

In[74]:=
```
f[c_] := σF/2 (Log[(c / a)^2] + 1 - (c / b)^2) - pp
```

In[75]:=
```
f[a]
```

Out[75]= -2.00988×10^8

In[76]:=
```
f[(a + b)/2]
```

Out[76]= 2.54985×10^8

Ezután az (a, $\frac{a+b}{2}$) intervallum további felezésével lehet a megoldás felé haladni:

In[77]:=
```
c3 = ((a+b)/2 + a)/2
```

Out[77]= 0.094375

In[78]:=
```
f[c3]
```

Out[78]= 9.04366×10^7

In[79]:=
```
c4 = (c3 + a)/2
```

Out[79]= 0.0784375

In[80]:=
```
f[c4]
```

Out[80]= -3.43666×10^7

Mivel $f(c_4)$ negatív lett, most a c_3 és c_4 intervallum felét választjuk

In[81]:=
```
c5 = (c3 + c4)/2
```

Out[81]= 0.0864062

In[82]:=
```
f[c5]
```

Out[82]= 3.24128×10^7

Végül a c_4 és c_5 intervallum közepét:

In[83]:=
```
c6 = (c4 + c5)/2
```

Out[83]= 0.0824219

In[84]:=
```
f[c6]
```

Out[84]= 204 376.

Ez az érték (0.2 MPa) már elegendően kicsi, elfogadhatjuk a c_6 = 82.42 mm értéket.

Megjegyzés: összesen 6-szor kellett kiszámolni a $\frac{\sigma_F}{2}\left(\ln\left[\left(\frac{c}{a}\right)^2\right] + 1 - \left(\frac{c}{b}\right)^2\right) - p$ függvény értékét.

c) Feszültségek a képlékeny zónában

$$\sigma_r^{\text{képl}}(r) = \sigma_F \ln\left(\frac{r}{a}\right) - p, \quad \sigma_\theta^{\text{képl}}(r) = \sigma_F\left(\ln\left(\frac{r}{a}\right) + 1\right) - p$$

In[85]:= $\sigma rk[r\_] := \sigma F \, Log\left[\frac{r}{a}\right] - pp$

$\sigma\theta k[r\_] := \sigma F \left(Log\left[\frac{r}{a}\right] + 1\right) - pp$

Feszültségeloszlások a képlékeny zónában.

In[87]:= Plot$\left[\{\sigma rk[r], \sigma\theta k[r]\}, \{r, a, c\}, PlotLegends \to \left\{"\sigma_r^{képl}(r)", "\sigma_\theta^{képl}(r)"\right\},\right.$
$\left. AxesLabel \to \{"r", "\sigma"\}, PlotRange \to \{\{0, b\}, All\}, Filling \to Axis\right]$

Out[87]=



In[88]:= $\sigma\theta k[c]$

Out[88]= 5.0493×10^{8}

Azaz 505 MPa

d) Egyenértékű maradó feszültség

Rugalmas tehermentesítést feltételezve a p_p nyomás esetén az A és B paraméterek:

In[89]:= $App = \frac{pp \, a^2}{b^2 - a^2}$ // N

Out[89]= 7.03747×10^{7}

In[90]:= $Bpp = (pp) \frac{a^2 \, b^2}{b^2 - a^2}$ // N

Out[90]= 2.54053×10^{6}

A tehermentesítés feszültségeloszlásaiban A_{pp} és B_{pp} paramétereket kell használni, és a feszültségek ellentétes előjelűek lesznek:

In[91]:= $\sigma rTM[r\_] := -\left(App - \frac{Bpp}{r^2}\right)$

$\sigma\theta TM[r\_] := -\left(App + \frac{Bpp}{r^2}\right)$

A maradó feszültségek a belső peremen

In[93]:= $\sigma rM = \sigma rk[a] + \sigma rTM[a]$

Out[93]= 0.

In[94]:=
```
σθM = σθk[a] + σθTM[a]
```

Out[94]= -4.50749×10^{8}

Ezek különbsége adja a Mohr-féle egyenértékű maradó feszültséget:

In[95]:=
```
σeM = Abs[σθM - σrM]
```

Out[95]= 4.50749×10^{8}

Azaz $\sigma^{M}_{e,\text{maradó}}(a) = 450.75\ \text{MPa}$